



DECUS

PROGRAM LIBRARY

DECUS NO.

8-201

TITLE

DECSW

AUTHOR

Kenneth B. Wiberg

COMPANY

Yale University
New Haven, Connecticut

DATE

June 2, 1969

SOURCE LANGUAGE


DECUS

STANDARD & BUREAU



This subroutine accepts the contents of decimal switches at a remote location, and converts the number into several forms: 1, as an insert into a BCD string which may be typed out or displayed on a CRT screen; 2, as a floating point number in the floating point accumulator; and 3, if an integer, as the binary equivalent in the accumulator.


Specifications: The subroutine assumes that a set of input switches are attached to the computer via an interface. The binary coded switches are arranged as follows:



The sign switch is EEC0 No. 165M (0 indicates + and 1 indicates -), the decimal switches are EEC0 No. 119M (1-2-4-8 output), and the decimal point location switch is EEC0 No. 148M (1-2-4-8 output plus a decimal, 0-9, output). The latter may be used to light pilot lights which indicate the location of the decimal point with respect to the decimal switches. An indication of 6 is arbitrarily taken as indicating the decimal point just to the right of the decimal switches, i. e., an integer. Smaller numbers shift the decimal point to the left. The

subroutine requires two pages in memory and uses the floating point interpreter.

Description: The contents of the decimal switches are read into the computer via IOT1 and IOT2 pulses (from a W103 device selector) and four R123 modules. The -3v source is supplied by W005 clamped loads. When read in as two words, the number appears as follows:



The decimal point is extracted from word 1, incremented and stored in DVAL. The integers in the number are then extracted in turn after checking for the decimal point. The BCD representations of the sign, the decimal point and four integers are packed two to a word into the three locations starting with the address LOC. At the same time, the integers are converted to binary form by multiplying the previously developed number by ten and adding the new integer. The final double precision binary number is converted to floating point form with the correct exponent by converting to a floating point integer and then multiplying or dividing by ten the required number of times. If the

number is an integer (i. e. the decimal point indicator was set to six), the number is also found in the accumulator on return from the subroutine.

The internal subroutine SHIFTS shifts the two word BCD input number in order to extract the individual decimal integers. MPY10S effects multiplication of the double precision binary number by ten. SIGNS sets the sign of the binary number after the BCD to binary conversion.

Usage: The subroutine is designed to be used with the Teletype output routine, TYPSTG (DIGITAL 8-20-U) or the CRT output routine, GIAN (DECUS 8-23). The address of the first of three successive locations into which the BCD representation is to be inserted is loaded into the accumulator before calling the subroutine. When called, the subroutine directs the computer to read the contents of the digital switches and effect the conversion. On return from the subroutine, the BCD representation is in the proper locations, the floating point representation is in the floating point interpreter (locations 44, 45 and 46), and if an integer, the binary representation is found in the accumulator.

Note: The subroutine may be checked independently of the switches by using the attached program which simulates the decimal switches using the console switch register.

The binary version of the test program is loaded after the DECSW subroutine, and version C of the floating point interpreter is also loaded. Start at location 200. The computer will halt. Word 1 (see pg. 2 of writeup) is formed using the switch register, and the Continue key is depressed. Another halt occurs, and word 2 is entered in the same fashion. The BCD representation of the number will be typed followed by the floating point form. Note that a decimal point is not printed if the proper position is not within the character string (ie. the number is too small or too large to be typed in fixed point form). Also, remember that a decimal point setting of six indicates the decimal point to follow the least significant integer.

/SUBROUTINE FOR READING CONTENTS OF DECIMAL SWITCHES
 /INPUT WITH LOCATION OF BCD STRING IN AC
 /OUTPUT IN FLOATING POINT ACCUMULATOR AND BCD STRING
 /ALSO IN AC IF INTEGER <2048

*3400

| | | | | |
|------|------|--------|--------------|------------------------------|
| 3400 | 0000 | DECSW, | 0 | |
| 3401 | 3363 | | DCA LOC | |
| 3402 | 3044 | | DCA 44 | /CLEAR FL PT AC |
| 3403 | 3045 | | DCA 45 | |
| 3404 | 3046 | | DCA 46 | |
| 3405 | 6371 | | 6371 | /READ UPPER HALF OF SWITCH |
| 3406 | 3353 | | DCA T1 | |
| 3407 | 6372 | | 6372 | /READ LOWER HALF OF SWITCH |
| 3410 | 3354 | | DCA T2 | |
| 3411 | 1353 | | TAD T1 | /LOCATE DECIMAL POINT |
| 3412 | 7012 | | RTR | |
| 3413 | 7012 | | RTR | |
| 3414 | 0355 | | AND MASK | |
| 3415 | 7041 | | CIA | |
| 3416 | 7001 | | IAC | |
| 3417 | 3361 | | DCA DVAL | |
| 3420 | 1361 | | TAD DVAL | |
| 3421 | 3364 | | DCA INDX | |
| 3422 | 1363 | | TAD LOC | /LOCATION OF BCD STRING |
| 3423 | 3365 | | DCA INDLOC | |
| 3424 | 1352 | | TAD M4 | |
| 3425 | 3367 | | DCA IN4 | /CHARACTER COUNTER |
| 3426 | 1351 | | TAD M2 | |
| 3427 | 3366 | | DCA IN2 | /BCD POSITION COUNTER |
| 3430 | 4776 | | CALL SIGN | /SET SIGN OF NUMBER |
| 3431 | 2364 | LOOP, | ISZ INDX | /DECIMAL POINT? |
| 3432 | 5241 | | JMP A2 | /NO |
| 3433 | 1362 | | TAD PER | /YES |
| 3434 | 2366 | | ISZ IN2 | /SECOND ENTRY IN WORD? |
| 3435 | 5240 | | JMP A1 | /NO |
| 3436 | 4774 | | CALL SHIFT | /YES |
| 3437 | 5241 | | JMP A2 | |
| 3440 | 3765 | A1, | DCA I INDLOC | |
| 3441 | 1353 | A2, | TAD T1 | /GET DECIMAL CHARACTER |
| 3442 | 0355 | | AND MASK | |
| 3443 | 3357 | | DCA TEMP | |
| 3444 | 1357 | | TAD TEMP | |
| 3445 | 1356 | | TAD SIXTY | /FORM BCD CHARACTER |
| 3446 | 2366 | | ISZ IN2 | /SECOND ENTRY? |
| 3447 | 5252 | | JMP B1 | /NO |
| 3450 | 4774 | | CALL SHIFT | /YES |
| 3451 | 5253 | | JMP B2 | |
| 3452 | 3765 | B1, | DCA I INDLOC | |
| 3453 | 1352 | B2, | TAD M4 | |
| 3454 | 3370 | | DCA INX | |
| 3455 | 1354 | LOOP2, | TAD T2 | /SHIFT DOUBLE PRECISION WORD |
| 3456 | 7104 | | CLL RAL | |
| 3457 | 3354 | | DCA T2 | |
| 3460 | 1353 | | TAD T1 | |

| | | | | |
|------|------|------|--------------|--------------------------------|
| 3461 | 7004 | | RAL | |
| 3462 | 3353 | | DCA T1 | |
| 3463 | 2370 | | ISZ INX | /SHIFT FINISHED? |
| 3464 | 5255 | | JMP LOOP2 | /CONTINUE LOOP |
| 3465 | 4775 | | CALL MPY10 | /FIX DECIMAL NUMBER |
| 3466 | 7100 | | CLL | |
| 3467 | 1046 | | TAD 46 | |
| 3470 | 1357 | | TAD TEMP | |
| 3471 | 3046 | | DCA 46 | |
| 3472 | 7430 | | SZL | |
| 3473 | 2045 | | ISZ 45 | |
| 3474 | 2367 | | ISZ IN4 | /ALL CHARACTERS DECODED? |
| 3475 | 5231 | | JMP LOOP | /NO: CONTINUE LOOP |
| 3476 | 2364 | | ISZ INDX | /FINAL DECIMAL POINT? |
| 3477 | 5306 | | JMP D2 | |
| 3500 | 1362 | | TAD PER | /YES |
| 3501 | 2366 | | ISZ IN2 | |
| 3502 | 5305 | | JMP D1 | |
| 3503 | 4774 | | CALL SHIFT | |
| 3504 | 5306 | | JMP D2 | |
| 3505 | 3765 | D1, | DCA I INDLOC | /SHIFT LAST CHARACTER? |
| 3506 | 2366 | D2, | ISZ IN2 | /NO |
| 3507 | 5311 | | JMP D3 | /YES |
| 3510 | 4774 | | CALL SHIFT | /CHECK SIGN OF NUMBER |
| 3511 | 4777 | D3, | CALL SIGN2 | |
| 3512 | 1046 | | TAD 46 | |
| 3513 | 3357 | | DCA TEMP | /SAVE LOW ORDER PART OF NUMBER |
| 3514 | 1360 | | TAD DCONST | /PUT IN FL. PT. FORM |
| 3515 | 3044 | | DCA 44 | |
| 3516 | 4407 | | FLPT | |
| 3517 | 7000 | | FNOR | |
| 3520 | 0000 | | FEXT | |
| 3521 | 1361 | | TAD DVAL | /MULT OR DIVIDE BY 10**I |
| 3522 | 1350 | | TAD CV | |
| 3523 | 7550 | | SPA SNA | |
| 3524 | 5335 | | JMP MPY | |
| 3525 | 7041 | | CIA | |
| 3526 | 3364 | | DCA INDX | /DIVIDE |
| 3527 | 4407 | | FLPT | |
| 3530 | 4371 | | FDIV TEN | |
| 3531 | 0000 | | FEXT | |
| 3532 | 2364 | | ISZ INDX | |
| 3533 | 5327 | | JMP .-4 | |
| 3534 | 5345 | | JMP C1 | |
| 3535 | 7500 | MPY, | SMA | |
| 3536 | 5346 | | JMP C2 | /MULTIPLY |
| 3537 | 3364 | | DCA INDX | |
| 3540 | 4407 | | FLPT | |
| 3541 | 3371 | | FMPY TEN | |
| 3542 | 0000 | | FEXT | |
| 3543 | 2364 | | ISZ INDX | |
| 3544 | 5340 | | JMP .-4 | |
| 3545 | 5600 | C1, | RETURN DECSW | |
| 3546 | 1357 | C2, | TAD TEMP | |

| | | | |
|------|------|---------|-----------|
| 3547 | 5345 | | JMP .-2 |
| 3550 | 0005 | CV, | 5 |
| 3551 | 7776 | M2, | -2 |
| 3552 | 7774 | M4, | -4 |
| 3553 | 0000 | T1, | 0 |
| 3554 | 0000 | T2, | 0 |
| 3555 | 0017 | MASK, | 0017 |
| 3556 | 0060 | SIXTY, | 0060 |
| 3557 | 0000 | TEMP, | 0 |
| 3560 | 0027 | DCONST, | 27 |
| 3561 | 0000 | DVAL, | 0 |
| 3562 | 0056 | PER, | 0056 |
| 3563 | 0000 | LOC, | 0 |
| 3564 | 0000 | INDX, | 0 |
| 3565 | 0000 | INDLOC, | 0 |
| 3566 | 0000 | IN2, | 0 |
| 3567 | 0000 | IN4, | 0 |
| 3570 | 0000 | INX, | 0 |
| 3571 | 0004 | TEN, | FLTG 10.0 |
| 3572 | 2377 | | |
| 3573 | 7776 | | |
| 3574 | 3600 | SHIFT, | SHIFTS |
| 3575 | 3623 | MPY10, | MPY10S |
| 3576 | 3662 | SIGN, | SIGNS |
| 3577 | 3712 | SIGN2, | SIGNS2 |

/DECSW SUBROUTINES

```

*3600
3600 0000 SHIFTS, 0
3601 3307 DCA TEMV
3602 1617 TAD I LOCV
3603 3220 DCA LOCS
3604 7100 CLL
3605 1620 TAD I LOCS
3606 7006 RTL
3607 7006 RTL
3610 7006 RTL
3611 1307 TAD TEMV
3612 3620 DCA I LOCS
3613 1221 TAD MM2
3614 3622 DCA I INN2
3615 2617 ISZ I LOCV
3616 5600 RETURN SHIFTS
3617 3565 LOCV, INDLOC
3620 0000 LOCS, 0
3621 7776 MM2, -2
3622 3566 INN2, IN2
3623 0000 MPY10S, 0
3624 7200 CLA
3625 1045 TAD 45
3626 3246 DCA T4
3627 1046 TAD 46
3630 3247 DCA T5
3631 4250 JMS SHFT2
3632 4250 JMS SHFT2
3633 7300 CLA CLL
3634 1046 TAD 46
3635 1247 TAD T5
3636 3046 DCA 46
3637 7430 SZL
3640 2045 ISZ 45
3641 1045 TAD 45
3642 1246 TAD T4
3643 3045 DCA 45
3644 4250 JMS SHFT2
3645 5623 RETURN MPY10S
3646 0000 T4, 0
3647 0000 T5, 0
3650 0000 SHFT2, 0
3651 7300 CLA CLL
3652 1046 TAD 46
3653 7004 RAL
3654 3046 DCA 46
3655 1045 TAD 45
3656 7004 RAL
3657 3045 DCA 45
3660 5650 RETURN SHFT2

```

| | | | | |
|------|------|---------|---------------|------------------------|
| 3661 | 3557 | TVAL, | TEMP | |
| 3662 | 0000 | SIGNS, | 0 | /SET BCD SIGN |
| 3663 | 1711 | | TAD I TT1 | |
| 3664 | 0310 | | AND MASK2 | |
| 3665 | 3307 | | DCA TEMV | |
| 3666 | 1617 | | TAD I LOCV | |
| 3667 | 3220 | | DCA LOCS | |
| 3670 | 3306 | | DCA FLAG | /CLEAR MINUS FLAG |
| 3671 | 1307 | | TAD TEMV | |
| 3672 | 7450 | | SNA | |
| 3673 | 5302 | | JMP PLUS | /PLUS |
| 3674 | 7300 | | CLA CLL | /MINUS |
| 3675 | 2306 | | ISZ FLAG | /SET MINUS FLAG |
| 3676 | 1304 | | TAD MINUS | /YES |
| 3677 | 3620 | | DCA I LOCS | |
| 3700 | 2622 | | ISZ I INN2 | |
| 3701 | 5662 | | RETURN SIGNS | |
| 3702 | 1305 | PLUS, | TAD PLUSS | |
| 3703 | 5277 | | JMP .-4 | |
| 3704 | 0055 | MINUS, | 0055 | |
| 3705 | 0053 | PLUSS, | 0053 | |
| 3706 | 0000 | FLAG, | 0 | |
| 3707 | 0000 | TEMV, | 0 | |
| 3710 | 7400 | MASK2, | 7400 | |
| 3711 | 3553 | TT1, | T1 | |
| 3712 | 0000 | SIGNS2, | 0 | |
| 3713 | 7200 | | CLA | |
| 3714 | 1306 | | TAD FLAG | /CHANGE SIGN OF FL AC? |
| 3715 | 7550 | | SPA SNA | |
| 3716 | 5712 | | RETURN SIGNS2 | /NO: EXIT |
| 3717 | 7300 | | CLA CLL | /YES |
| 3720 | 1046 | | TAD 46 | |
| 3721 | 7041 | | CIA | |
| 3722 | 3046 | | DCA 46 | |
| 3723 | 1045 | | TAD 45 | |
| 3724 | 7040 | | CMA | |
| 3725 | 7430 | | SZL | |
| 3726 | 7101 | | CLL IAC | |
| 3727 | 3045 | | DCA 45 | |
| 3730 | 5712 | | RETURN SIGNS2 | |

| | |
|--------|------|
| A1 | 3440 |
| A2 | 3441 |
| B1 | 3452 |
| B2 | 3453 |
| CV | 3550 |
| C1 | 3545 |
| C2 | 3546 |
| DCONST | 3560 |
| DECSW | 3400 |
| DVAL | 3561 |
| D1 | 3505 |
| D2 | 3506 |
| D3 | 3511 |
| FLAG | 3706 |
| INDLOC | 3565 |
| INDX | 3564 |
| INN2 | 3622 |
| INX | 3570 |
| IN2 | 3566 |
| IN4 | 3567 |
| LOC | 3563 |
| LOCS | 3620 |
| LOCV | 3617 |
| LOOP | 3431 |
| LOOP2 | 3455 |
| MASK | 3555 |
| MASK2 | 3710 |
| MINUS | 3704 |
| MM2 | 3621 |
| MPY | 3535 |
| MPY10 | 3575 |
| MPY10S | 3623 |
| M2 | 3551 |
| M4 | 3552 |
| PER | 3562 |
| PLUS | 3702 |
| PLUSS | 3705 |
| SHFT2 | 3650 |
| SHIFT | 3574 |
| SHIFTS | 3600 |
| SIGN | 3576 |
| SIGNS | 3662 |
| SIGNS2 | 3712 |
| SIGN2 | 3577 |
| SIXTY | 3556 |
| TEMP | 3557 |
| TEMV | 3707 |
| TEN | 3571 |
| TT1 | 3711 |
| TVAL | 3661 |
| T1 | 3553 |
| T2 | 3554 |
| T4 | 3646 |
| T5 | 3647 |

/DECSW CHECKING PROGRAM

*200

```

0200 6046 TLS
0201 7200 CLA
0202 7402 HLT
0203 7404 OSR
0204 3777 DCA T1
0205 7402 HLT
0206 7404 OSR
0207 3776 DCA T2
0210 1252 TAD LOCATN
0211 4656 CALL DEC
0212 4407 FLPT
0213 0014 OUTPUT
0214 0000 FEXT
0215 7200 CLA
0216 1253 TAD L1
0217 4227 JMS PRT
0220 7200 CLA
0221 1254 TAD L2
0222 4227 JMS PRT
0223 7200 CLA
0224 1255 TAD L3
0225 4227 JMS PRT
0226 5201 JMP 201
0227 0000 PRT,0
0230 3260 DCA TEMPZ
0231 1260 TAD TEMPZ
0232 7012 RTR
0233 7012 RTR
0234 7012 RTR
0235 0257 AND MASKS
0236 4244 JMS PTS
0237 7200 CLA
0240 1260 TAD TEMPZ
0241 0257 AND MASKS
0242 4244 JMS PTS
0243 5627 RETURN PRT
0244 0000 PTS,0
0245 1261 TAD X
0246 6041 TSF
0247 5246 JMP .-1
0250 6046 TLS
0251 5644 RETURN PTS
0252 0253 LOCATN,L1
0253 0000 L1,0
0254 0000 L2,0
0255 0000 L3,0
0256 3400 DEC,3400
0257 0077 MASKS,0077
0260 0000 TEMPZ,0
0261 0200 X,200
0376 3554 *3405
0377 3553
3405 7000 NOP
3406 7000 NOP
3407 7000 NOP
3410 7000 NOP
      *3553
3553 0000 T1,0
3554 0000 T2,0

```

